

## SiC Power Devices

# How to Use Thermal Models

Among SPICE models, there are models for performing simulations in relation to heat, which are referred to as thermal models. Simulations using the thermal models are performed to make a rough estimate during the initial stage of thermal design. This application note explains how to use the thermal models.

## Obtaining thermal models

Thermal models are available from ROHM's website. From the search window, search for the model name and click on the applicable model name in the results. This will take you to the individual product page, where you can select the design model in the "TOOLS" tab and download the thermal calculation model.

## Overview of thermal models

Figure 1 shows an example of a thermal model. The example is for a 1,200V Nch SiC MOSFET SCT3040KR, and the same diagram is also shown on the data sheet. The Tj pin shows the junction temperature, the Tc pin shows the case temperature, and the Ta pin shows the ambient temperature.




Figure 1. Example of thermal model: Cauer RC thermal circuit network

SiC MOSFET manufactured by ROHM: SCT3040KR

Figure 3 shows the netlist for the thermal model that was downloaded. Both the electrical and thermal models for the SiC MOSFET are shown in this netlist. The chip and pin configuration is as shown in the circuit diagram of Figure 2.

Further, this model includes a function to transfer the power

consumption (self-heating) generated by the SiC MOSFET electrical model to the thermal model. For this reason, voltage equivalent to the junction temperature is generated in the Tj pin when the electrical circuit simulation is performed in the usual way.

Now, let's take a brief look at the contents of the netlist. To explain the function for transferring the power consumption that occurs on the SiC MOSFET to the thermal model, we have extracted the circuits for only the major parts of the netlist, as shown in Figure 4. The diagram is broken into three blocks. Block A is for the SiC MOSFET; block B is for the body diode; and block C is for the thermal resistance model.

Comparing with the netlist on Figure 3, we can see that block A corresponds with lines 13 through 42; block B corresponds with lines 44 through 53; and block C corresponds with lines 55 through 62. E20 on line 55 (voltage-controlled voltage source) and R20 on line 56 (resistor) are components used to transfer the power consumption of the SiC MOSFET and body diode to the Tj pin of the thermal resistance model as the current amount. Normally, the current source is used to supply current value corresponding to the power consumption to the Tj pin. However, as the output impedance of the current source is high and the convergence for the initial value may not be good, the current source for low output impedance is expressed as the combination of the voltage source and output resistance. The 1kΩ value of output impedance R20 is a sufficiently large value for thermal resistance.




Figure 2. Chip and pin configuration included in thermal calculation model

SiC MOSFET manufactured by ROHM: SCT3040KR

Next, we will explain the detectable parts of the power consumption that occurs in the SiC MOSFET and body diode. This is done using the calculation formula listed in E20, on line 55 of the netlist. We have extracted this part for formula (1).

$$P_D = (I(V1) - I(V11)) \times V(1,3) + I(V2) \times V(2,3) \quad (1)$$

```

1 | * SCT3040KR_T
2 | * SiC NMOSFET with driver source Self-heating Thermal model
3 | * 1200V 55A 40mOhm
4 | * Model Generated by ROHM
5 | * All Rights Reserved
6 | * Commercial Use or Resale Restricted
7 | * Date: 2019/07/09
8 | *****D G S DS Tj Tc Ta
9 | SUBCKT SCT3040KR_T 1 2 3 4 Tj Tc Ta
10 | .PARAM TO=25 T1=-100 T2=300
11 | .FUNC K1 (T)      {MIN (MAX (T, T1), T2)}
12 | *****
13 | .FUNC R1 (I, T)   {30.33m*I*EXP ((T-T0)/166.5)+4.822u*I*ABS (I)**1.596*EXP ((T-T0)/892.6)}
14 | .FUNC R2 (I)      {1*(1.3*(1+TANH (I))/2+3.7*(1-TANH (I))/2)}
15 | .FUNC V1 (I, T)   {139.4m*ASINH (I/0.1277)*EXP ((T-T0)/-409.0)+
16 | +                559.5m*ASINH (I/3.997)*EXP ((T-T0)/-1244)+25.41m*I*EXP ((T-T0)/498.5)}
17 | .FUNC V2 (V, T)   {21.45p*V**13.56*EXP ((T-T0)/38.52*EXP ((T-T0)/-1793)}
18 | .FUNC V3 (V, T)   {94.67*SINH (V/94.67)*EXP ((T-T0)/11811)}
19 | .FUNC I1 (V, W, T) {V3 (V, T)*(1+3.258*(TANH ((V3 (V, T)-66.90)/113.2)+1)*EXP ((T-T0)/-174.1)/10)*
20 | +                W/(ABS (W)+3.258*(TANH ((V3 (V, T)-66.90)/113.2)+1)*EXP ((T-T0)/-174.1))}
21 | .FUNC C1 (U, V, W) {(559.5*V+1446*(1-576.3*TANH (W/576.3)/1.345)**-0.5204)*(1+(TANH ((U+8.327)/3.706)-1)/5.183)}
22 | V1 1 11 0
23 | E1 11 12 VALUE={R1 (I (V1), K1 (V (Tj)))}
24 | V2 2 21 0
25 | E2 21 22 VALUE={R2 (I (V2))}
26 | L1 3 32 7.0n
27 | R2 3 32 10
28 | R5 4 32 1m
29 | E6 41 0 VALUE={MIN (MAX (V (22, 32), 0), 22)}
30 | V6 41 42 0
31 | E7 42 43 VALUE={V1 (I (V6), K1 (V (Tj)))}
32 | G6 43 0 VALUE={MIN (MAX (V2 (MAX (V (43), 0), K1 (V (Tj))), -500), 500)}
33 | C6 43 0 1p
34 | R6 43 0 1T
35 | G1 12 32 VALUE={MIN (MAX (I1 (MAX (I (V6), 0), V (12, 32), K1 (V (Tj))), -300), 300)}
36 | C7 12 32 1p
37 | R3 12 32 1T
38 | V3 22 23 0
39 | C1 23 12 1p
40 | G2 22 12 VALUE={I (V3)*C1 (V (22, 12), MIN (MAX (V (22, 12), 0), 0.525), MIN (V (22, 12), 0))}
41 | C2 22 32 1.531n
42 | R4 22 3 1G
43 | *****
44 | .FUNC R11 (I, T) {87.51m*ASINH (I/0.4297)*EXP ((T-T0)/304.7)+43.40m*I*EXP ((T-T0)/303.9)}
45 | .FUNC I11 (V, T) {17.41n*(EXP (V/0.09161/EXP ((T-T0)/-311.1))-1)*EXP ((T-T0)/-64.07*EXP ((T-T0)/761.8))-
46 | +                1u*TANH (-V/0.1)*EXP ((T-T0)/180.3)-8.756E-30*(EXP (-V/25)-1)*EXP ((T-T0)/-200)}
47 | .FUNC C11 (V, W) {832.8*(V-1.091)+1065*(1-596.7*TANH (W/596.7)/2.183)**-0.5737}
48 | V11 32 51 0
49 | E11 51 52 VALUE={R11 (I (V11), K1 (V (Tj)))}
50 | V12 52 53 0
51 | C11 53 1 1p
52 | G11 52 1 VALUE={MIN (MAX (I11 (V (52, 1), K1 (V (Tj))), -300), 300)+I (V12)*C11 (MAX (V (52, 1), 1.091), MIN (V (52, 1), 1.091))}
53 | R11 52 1 1T
54 | *****
55 | E20 T0 Ta VALUE={1k*MIN (MAX (IF (TIME>0, (I (V1)-I (V11))*V (1, 3)+I (V2)*V (2, 3), 0), -2MEG), 2MEG)}
56 | R20 T0 Tj 1k
57 | C21 Tj Ta 1.385m
58 | C22 T2 Ta 10.02m
59 | C23 T3 Ta 35.72m
60 | R21 Tj T2 25.59m
61 | R22 T2 T3 194.8m
62 | R23 T3 Tc 219.6m
63 | .ENDS SCT3040KR_T

```

Figure 3. Example of a thermal model netlist  
SiC MOSFET manufactured by ROHM: SCT3040KR




Figure 4. Circuit with only the major parts of the netlist, used to explain the function for transferring the power consumption that occurs in the SiC MOSFET to the thermal model  
 SiC MOSFET manufactured by ROHM: SCT3040KR

For the first term,  $I(V1)$  is the current that flows between the drain and source of SiC MOSFET, and  $I(V11)$  is the current that flows in the body diode. Added current value (pay attention to the polarity of V11) is multiplied by the voltage value  $V(1,3)$  between the drain and source to calculate the power consumption between the drain and source. For the second term,  $I(V2)$  is the current flowing between gate and source; and  $V(2,3)$  is the voltage between gate and source. The loss between gate and source is calculated by multiplying these two terms. The overall power consumption is detected by adding the values of terms 1 and 2 together.

## Overview of simulation circuit

Figure 5 shows a typical simulation circuit that uses a thermal model. The device configuration at that time is shown in Figure 6. By applying ambient temperature  $T_A$  to the Ta pin as bias

voltage  $V_{BIAS}$ , and applying the device's power consumption  $P_D$  to the Tj pin as current I, a voltage with an RC time constant is generated on the Tj pin. This generated voltage represents the junction temperature. Further, for the resistance connected to the Tc pin, R1 is the thermal resistance  $R_{thCF}$  between the case and the heat sink, and R2 is the thermal resistance  $R_{thFA}$  between the heat sink and the ambient temperature. Note that  $R_{thCF}$  includes the thermal resistance of the thermal interface material (TIM) and the contact thermal resistance. Also, C1 is the heat capacity of the heat sink, and R2 and C1 comprise the heat sink.




Figure 5. Example of simulation circuit




Figure 6. Device configuration

## Simulation method

The setting method differs depending on the simulation used. This explanation uses SystemVision® Cloud (Mentor Graphics Corp.).




Figure 7. Creating circuit symbols for the thermal model

First, create the circuit symbols while referring to the netlist. The pin information is found on lines 8 through 9 of the netlist. In this example, pin numbers 1, 2, 3, and 4 are assigned to D (Drain), G (Gate), S (Source) and DS (Driver Source) for the electrical model in that order. Further, pin numbers Tj, Tc and Ta are assigned to thermal model pins Tj, Tc and Ta, making circuit

symbols for a total of 7 pins. Use whatever design you like for the symbols, as long as the seven pins are in order. Figure 7 shows an example of how this is made. Although there are two components, this is represented by one symbol, as the simulation circuit is separated to make it easier to write.

Since the netlist for the thermal model is listed for PSpice, you will get a syntax error when used with SystemVision® Cloud. On this example, the IF statement on line 55 of the netlist needs to be changed to VALIF, as shown below.

### Source

```
E20 T0 Ta VALUE={1k*MIN(MAX(IF(TIME>0,(I(V1)-I(V11))*V(1,3)+I(V2)*V(2,3),0),-2MEG),2MEG)}
```

### After modification

```
E20 T0 Ta VALUE={1k*MIN(MAX(VALIF(TIME>0,(I(V1)-I(V11))*V(1,3)+I(V2)*V(2,3),0),-2MEG),2MEG)}
```

Figure 8 shows the simulation circuit. Here, we will perform a simulation assuming that a heat sink is mounted onto TO-247-4L (Figure 9) of the through-hole package. The left half of the circuit diagram is the electrical circuit, and the right half is the thermal model circuit. We will calculate the junction temperature Tj for a SiC MOSFET: SCT3040KR.

A square wave is inputted to the SiC MOSFET gate from signal source vsrc2 and the switching operation is performed on the electrical circuit. 600 V DC power is supplied from vsrc1 and connected to the SiC MOSFET drain via the 45Ω load resistor r1. Once the SiC MOSFET turns on, approximately 13 A of drain current will flow. At that time, conductive loss will occur due to the conductive resistance of the SiC MOSFET and the drain current. The voltage waveforms of the gate and drain for the SiC MOSFET is shown in Figure 10.

The voltage source vsrc3 for the ambient temperature is connected to the Ta pin of the thermal model circuit, and voltage (1°C=1 V) is applied. As the Tc pin is a package case, the connection to the Ta pin is made through r4 (meaning the thermal resistance RthCF between the case and the heat sink) and r5 (meaning the thermal resistance RthFA between the heat sink and the ambient temperature). For r4, silicon grease or the likes is assumed to be used between the case and heat sink. r5 is the thermal resistance for the heat sink. The values for both

r4 and r5 are applied at the resistor value ( $1^{\circ}\text{C}/\text{W}=1\Omega$ ). c1 is the heat capacity of the heat sink, and connects to the Ta pin. r5 and c1 comprise the heat sink. Refer either to the heat sink data sheet or the estimated value using formula (2) for the heat capacity value of c1. The value of c1 is applied using the capacitance ( $1\text{ J/K}=1\text{ F}$ ).

$$\text{Heat capacity } C = c \times \rho \times V \quad [\text{J/K}] \quad (2)$$

c : Specific heat capacity  $[\text{kJ/kg} \cdot \text{K}]$

$\rho$  : Density  $[\text{g/cm}^3]$

V : Volume  $[\text{cm}^3]$

For example, in a case of an aluminum heat sink whose shape is a  $3\text{ cm} \times 3\text{ cm} \times 1\text{ cm}$  cuboid, use the following formula to calculate.

Specific heat capacity of aluminum  $c = 0.896 [\text{kJ/kg} \cdot \text{K}]$

Density of aluminum  $\rho = 2.71 [\text{g/cm}^3]$

Volume  $V = 3 \times 3 \times 1 = 9 [\text{cm}^3]$

$$C = 0.896 \times 2.71 \times 9 = 21.8 [\text{J/K}] = 21.8 [\text{F}] \quad (3)$$

To get the junction temperature (meaning the voltage at the Tj point), the power consumption generated in the SiC MOSFET needs to be converted to current and applied to the Tj pin of the thermal model; but as previously explained, this is done through the calculation formula (1) within the thermal model. Note that the resistor r3 that is connected between the Tj pin and the ground is used for termination processing at a resistance value that will not have an effect on the calculation, as errors will occur during simulation if the Tj pin is open.

Lastly, perform a transient analysis and monitor the voltage of the Tj point to determine the junction temperature ( $1\text{ V}=1^{\circ}\text{C}$ ).




Figure 8. Simulation circuit



Figure 9. TO-247-4L package




Figure 10. Voltage waveform of the x1 gate (below) and drain (above)

An example of implementation is shown below. When the chip junction temperature reaches 100°C, the circuit will stop operating. In consideration of the integration of a circuit to keep down the increase in temperature, we perform a simulation to find how much time the circuit can be operated.

For the simulation parameters, an aluminum heat sink with dimensions of 3 cm × 3 cm × 1 cm is mounted. We assume that the thermal resistance for r4 is 0.1 W/°C due to the silicon grease

between the case and heat sink, and that the thermal resistance for r5 due to the heat sink thermal resistance is 15°C/W. c1 is the heat capacity of the heat sink, and is set with the value calculated in the previous formula (3). The ambient temperature is 50°C.

The result is shown in Figure 11. We can see that the junction temperature reaches 100°C around 350 seconds after the circuit begins operating.




Figure 11. Results of transient analysis, of the time for junction temperature to reach 100°C with heat sink mounted

Assumptions: thermal resistance 15°C/W, aluminum heat sink of 3 cm × 3 cm × 1 cm in size

SiC MOSFET manufactured by ROHM: SCT3040KR

SystemVision® is a registered trademark of Mentor Graphics Corp.



## Notes

- 1) The information contained herein is subject to change without notice.
- 2) Before you use our Products, please contact our sales representative and verify the latest specifications :
- 3) Although ROHM is continuously working to improve product reliability and quality, semiconductors can break down and malfunction due to various factors.  
Therefore, in order to prevent personal injury or fire arising from failure, please take safety measures such as complying with the derating characteristics, implementing redundant and fire prevention designs, and utilizing backups and fail-safe procedures. ROHM shall have no responsibility for any damages arising out of the use of our Products beyond the rating specified by ROHM.
- 4) Examples of application circuits, circuit constants and any other information contained herein are provided only to illustrate the standard usage and operations of the Products. The peripheral conditions must be taken into account when designing circuits for mass production.
- 5) The technical information specified herein is intended only to show the typical functions of and examples of application circuits for the Products. ROHM does not grant you, explicitly or implicitly, any license to use or exercise intellectual property or other rights held by ROHM or any other parties. ROHM shall have no responsibility whatsoever for any dispute arising out of the use of such technical information.
- 6) The Products specified in this document are not designed to be radiation tolerant.
- 7) For use of our Products in applications requiring a high degree of reliability (as exemplified below), please contact and consult with a ROHM representative : transportation equipment (i.e. cars, ships, trains), primary communication equipment, traffic lights, fire/crime prevention, safety equipment, medical systems, servers, solar cells, and power transmission systems.
- 8) Do not use our Products in applications requiring extremely high reliability, such as aerospace equipment, nuclear power control systems, and submarine repeaters.
- 9) ROHM shall have no responsibility for any damages or injury arising from non-compliance with the recommended usage conditions and specifications contained herein.
- 10) ROHM has used reasonable care to ensure the accuracy of the information contained in this document. However, ROHM does not warrants that such information is error-free, and ROHM shall have no responsibility for any damages arising from any inaccuracy or misprint of such information.
- 11) Please use the Products in accordance with any applicable environmental laws and regulations, such as the RoHS Directive. For more details, including RoHS compatibility, please contact a ROHM sales office. ROHM shall have no responsibility for any damages or losses resulting non-compliance with any applicable laws or regulations.
- 12) When providing our Products and technologies contained in this document to other countries, you must abide by the procedures and provisions stipulated in all applicable export laws and regulations, including without limitation the US Export Administration Regulations and the Foreign Exchange and Foreign Trade Act.
- 13) This document, in part or in whole, may not be reprinted or reproduced without prior consent of ROHM.



Thank you for your accessing to ROHM product informations.  
More detail product informations and catalogs are available, please contact us.

## ROHM Customer Support System

<http://www.rohm.com/contact/>